Colours missing, esp. green.

Spell check!

Meta–Groups, Meta–Bicrossed–Products, and the Alexander Polynomial, 1

Dror Bar–Natan at Knots in Washington XXXIV
http://www.math.toronto.edu/~drorbn/Talks/GWU-1203/

Abstract. A straightforward proposal for a group-theoretic invariant of knots fails if one really means groups, but works once generalized to meta-groups (to be defined). We will construct one complicated but elementary meta-group as a meta-bicrossed-product (to be defined), and explain how the resulting invariant is a not-yet-understood generalization of the Alexander polynomial, while at the same time being a specialization of a somewhat-understood "universal finite type invariant of w-knots" and of an elusive "universal finite type invariant of v-knots".





"Divide and Conquer"



Idea. Given a group G and two pairs $R^\pm = (g_o^\pm, g_u^\pm) \in G^2$, map them to xings and "multiply along", so that



$$\overset{Z}{\longmapsto} \left(\begin{array}{c} g_o^+ g_u^+ g_o^+ g_u^- g_o^- g_u^+ g_o^+ g_u^+ \\ g_u^- g_o^- \end{array} \right)$$

This Fails! R2 implies that $g_o^\pm g_u^\mp = e$ and then R3 implies that g_o^+ and g_u^+ commute, so the result is a simple counting invariant.

A Group Computer. Given G, can store group elements and perform operations on them:



(Also has S_x for inversion, e_x for unit insertion, d_x for register deletion, Δ^z_{xy} for element cloning, and $(D_1, D_2) \mapsto D_1 \cup D_2$ for merging, and very many obvious composition axioms relating these.

A Meta-Group. Is a similar "computer", only its internal structure is unknown to us. Namely it is a collection of sets $\{G_X\}$ indexed by all finite sets X, and a collection of operations m_z^{xy}, S_x, e_x, d_x, Δ^z_{xy}, and $\cup$, satisfying the exact same properties.

Example 1. The non-meta example, $G_X := G^X$.

Example 2. $G_X := M_{X\times X}(\mathbb{Z})$, with simultaneous row and column operations.

Bicrossed Prodcuts. If $G = HT$ is a group presented as a product of two of its subgroups, with $H \cap T = \{e\}$, then also $G = TH$ and G is determined by H, T, and the "swap" map $sw^{th} : (t,h) \mapsto (h',t')$ defined by $th = h't'$. The map sw satisfies (1) and (2) below; conversely, if $sw : T\times H \to H \times T$ satisfies (1) and (2) (+ lesser conditions), then (3) defines a group structure on $H \times T$, the "bicrossed product".



A **Meta-Bicrossed-Product** is a collection of sets $\beta(H,T)$ and operations tm_z^{xy}, hm_z^{xy} and sw_{xy}^{th} (and lesser ones), such that tm and hm are "associative" and (1) and (2) hold (+ lesser conditions). A meta-bicrossed-product defines a meta-group with $G_X := \beta(X,X)$ and gm as in (3).

β Calculus. Let $\beta(H,T)$ be

$$\left\{ \begin{array}{c|ccc} \omega & h_1 & h_2 & \cdots \\ \hline t_1 & \alpha_{11} & \alpha_{12} & \cdot \\ t_2 & \alpha_{21} & \alpha_{22} & \cdot \\ \vdots & \cdot & \cdot & \cdot \end{array} \;\middle|\; \begin{array}{l} h_j \in H,\ t_i \in T, \text{ and } \omega \text{ and} \\ \text{the } \alpha_{ij} \text{ are Laurent poly-} \\ \text{nomials in variables } T_i, \text{ in} \\ \text{bijection with the } t_i\text{'s} \end{array} \right\},$$

with operations

$$tm_z^{xy} : \begin{array}{c|c} \omega & \cdots \\ \hline t_x & \alpha \\ t_y & \beta \\ \vdots & \gamma \end{array} \mapsto \begin{array}{c|c} \omega & \cdots \\ \hline t_z & \alpha+\beta \\ \vdots & \gamma \end{array},$$

$$hm_z^{xy} : \begin{array}{c|ccc} \omega & h_x & h_y & \cdots \\ \hline \vdots & \alpha & \beta & \gamma \end{array} \mapsto \begin{array}{c|cc} \omega & h_z & \cdots \\ \hline \vdots & \alpha+\beta+\langle\alpha\rangle\beta & \gamma \end{array},$$

$$sw_{xy}^{th} : \begin{array}{c|cc} \omega & h_y & \cdots \\ \hline t_x & \alpha & \beta \\ \vdots & \gamma & \delta \end{array} \mapsto \begin{array}{c|cc} \omega\epsilon & h_y & \cdots \\ \hline t_x & \alpha(1+\langle\gamma\rangle/\epsilon) & \beta(1+\langle\gamma\rangle/\epsilon) \\ \vdots & \gamma/\epsilon & \delta-\gamma\beta/\epsilon \end{array},$$

where $\epsilon := 1+\alpha$, $\langle\alpha\rangle := \sum_i \alpha_i$, and $\langle\gamma\rangle := \sum_{i\neq x}\gamma_i$, and let

$$R^p_{xy} := \begin{array}{c|cc} 1 & h_x & h_y \\ \hline t_x & 0 & T_x - 1 \\ t_y & 0 & 0 \end{array} \qquad R^m_{xy} := \begin{array}{c|cc} 1 & h_x & h_y \\ \hline t_x & 0 & T_x^{-1} - 1 \\ t_y & 0 & 0 \end{array}.$$

Theorem. Z^β is a tangle invariant (and much more). Restricted to knots, the ω part is the Alexander polynomial. Restricted to links, it contains the multivariable Alexander polynomial. Restricted to braids, it is equivalent to the Burau representation.

Why Happy? • Applications to w-knots. • Everything that I know about the Alexander polynomial can be expressed cleanly in this language (even if without proof), except HF, but including genus, ribbonness, cabling, v-knots, knotted graphs, etc., and there's potential for vast generalizations.

* Fits on one sheet, including implementation

Meta–Groups, Meta–Bicrossed–Products, and the Alexander Polynomial, 2

I mean business! `<< Utilities.m`

The key implementation trick is the bijection

$$\begin{array}{c|c} \omega & h_j \\ \hline t_i & \alpha_{ij} \end{array} \longleftrightarrow B(\omega, \sum_{i,j} \alpha_{ij} t_i h_j):$$

```
⟨μ_⟩ := μ /. t_ → 1;
tm_{x_,y_→z_}[β_] := β /. {t_{x|y} → t_z, T_{x|y} → T_z};
hm_{x_,y_→z_}[B[ω_, Λ_]] := Module[
  {α = D[Λ, h_x], β = D[Λ, h_y], γ = Λ /. h_{x|y} → 0},
  B[ω, (α + (1 + ⟨α⟩) β) h_z + γ] // βCollect];
sw_{x_,y_}[B[ω_, Λ_]] := Module[{α, β, γ, δ, ϵ},
  α = Coefficient[Λ, h_y t_x]; β = D[Λ, t_x] /. h_y → 0;
  γ = D[Λ, h_y] /. t_x → 0;    δ = Λ /. h_y | t_x → 0;
  ϵ = 1 + α;
  B[ω * ϵ, α (1 + ⟨γ⟩ / ϵ) h_y t_x + β (1 + ⟨γ⟩ / ϵ) t_x
     + γ / ϵ h_y                  + δ - γ * β / ϵ
  ] // βCollect];
gm_{x_,y_→z_}[β_] := β // sw_{x,y} // hm_{x,y→z} // tm_{x,y→z};
B /: B[ω1_, Λ1_] B[ω2_, Λ2_] := B[ω1 * ω2, Λ1 + Λ2];
Rp_{x_,y_} := B[1, (T_x - 1) t_x h_y];
Rm_{x_,y_} := B[1, (T_x^-1 - 1) t_x h_y];
```

```
{β = B[ω, Sum[α_{10 i+j} t_i h_j, {i, {1, 2, 3}}, {j, {4, 5}}]],
  β // tm_{1,2→1} // sw_{1,4},
  β // sw_{2,4} // sw_{1,4} // tm_{1,2→1}
} // ColumnForm
```

Some testing. . .



$$\begin{pmatrix} \omega & h_4 & h_5 \\ t_1 & \alpha_{14} & \alpha_{15} \\ t_2 & \alpha_{24} & \alpha_{25} \\ t_3 & \alpha_{34} & \alpha_{35} \end{pmatrix}$$

$$\begin{pmatrix} \omega(1+\alpha_{14}+\alpha_{24}) & h_4 & h_5 \\ t_1 & \frac{(\alpha_{14}+\alpha_{24})(1+\alpha_{14}+\alpha_{24}+\alpha_{34})}{1+\alpha_{14}+\alpha_{24}} & \frac{(\alpha_{15}+\alpha_{25})(1+\alpha_{14}+\alpha_{24}+\alpha_{34})}{1+\alpha_{14}+\alpha_{24}} \\ t_3 & \frac{\alpha_{34}}{1+\alpha_{14}+\alpha_{24}} & \frac{-\alpha_{15}\alpha_{34}-\alpha_{25}\alpha_{34}+\alpha_{35}+\alpha_{14}\alpha_{35}+\alpha_{24}\alpha_{35}}{1+\alpha_{14}+\alpha_{24}} \end{pmatrix}$$

$$\begin{pmatrix} \omega(1+\alpha_{14}+\alpha_{24}) & h_4 & h_5 \\ t_1 & \frac{(\alpha_{14}+\alpha_{24})(1+\alpha_{14}+\alpha_{24}+\alpha_{34})}{1+\alpha_{14}+\alpha_{24}} & \frac{(\alpha_{15}+\alpha_{25})(1+\alpha_{14}+\alpha_{24}+\alpha_{34})}{1+\alpha_{14}+\alpha_{24}} \\ t_3 & \frac{\alpha_{34}}{1+\alpha_{14}+\alpha_{24}} & \frac{-\alpha_{15}\alpha_{34}-\alpha_{25}\alpha_{34}+\alpha_{35}+\alpha_{14}\alpha_{35}+\alpha_{24}\alpha_{35}}{1+\alpha_{14}+\alpha_{24}} \end{pmatrix}$$


```
{Rm_{5,1} Rm_{6,2} Rp_{3,4} // gm_{1,4→1} // gm_{2,5→2} // gm_{3,6→3},
  Rp_{6,1} Rm_{2,4} Rm_{3,5} // gm_{1,4→1} // gm_{2,5→2} // gm_{3,6→3}}
```

$$\left\{\begin{pmatrix} 1 & h_1 & h_2 \\ t_2 & -\frac{-1+T_2}{T_2} & 0 \\ t_3 & \frac{-1+T_3}{T_2} & -\frac{-1+T_3}{T_3} \end{pmatrix}, \begin{pmatrix} 1 & h_1 & h_2 \\ t_2 & -\frac{-1+T_2}{T_2} & 0 \\ t_3 & \frac{-1+T_3}{T_2} & -\frac{-1+T_3}{T_3} \end{pmatrix}\right\}$$

Divide and conquer!

```
β = Rm_{12,1} Rm_{2,7} Rm_{8,3} Rm_{4,11} Rp_{16,5} Rp_{6,13} Rp_{14,9} Rp_{10,15}
```

8_{17}

1	h_1	h_3	h_5	h_7	h_9	h_{11}	h_{13}	h_{15}
t_2	0	0	0	$-\frac{-1+T_2}{T_2}$	0	0	0	0
t_4	0	0	0	0	0	$-\frac{-1+T_4}{T_4}$	0	0
t_6	0	0	0	0	0	0	$-1+T_6$	0
t_8	0	$-\frac{-1+T_8}{T_8}$	0	0	0	0	0	0
t_{10}	0	0	0	0	0	0	0	$-1+T_{10}$
t_{12}	$-\frac{-1+T_{12}}{T_{12}}$	0	0	0	0	0	0	0
t_{14}	0	0	0	0	$-1+T_{14}$	0	0	0
t_{16}	0	0	$-1+T_{16}$	0	0	0	0	0

Propaganda

```
Do[β = β // gm_{1,k→1}, {k, 2, 10}]; β
```

8_{17}, cont.

$\frac{T_1^2-T_{16}-T_1T_{16}}{T_1^2}$	h_1	h_{11}	h_{13}	h_{15}
t_1	$-\frac{(-1+T_1)T_{14}(T_1^2+T_{16}^2)}{T_1^2T_{12}(T_1^2+T_{16}-T_1T_{16})}$	$-\frac{(-1+T_1)(1-T_1-T_1^2)T_{14}T_{16}}{T_1(T_1^2+T_{16}-T_1T_{16})}$	$\frac{(-1+T_1)(1-T_1-T_1^2)T_{14}}{T_1^2-T_{16}-T_1T_{16}}$	$-1+T_1$
t_{12}	$-\frac{-1+T_{12}}{T_{12}}$	0	0	0
t_{14}	$\frac{(-1+T_{14})(-T_1+T_1^2-T_{16})}{T_{12}(T_1^2-T_{16}-T_1T_{16})}$	$\frac{(-1+T_1)(1-T_1+T_1^2)(-1+T_{14})T_{16}}{T_1(T_1^2-T_{16}-T_1T_{16})}$	$-\frac{(-1+T_1)(1-T_1+T_1^2)(-1+T_{14})}{T_1^2-T_{16}-T_1T_{16}}$	0
t_{16}	$\frac{T_1(-1+T_{16})}{T_{12}(T_1^2-T_{16}-T_1T_{16})}$	$\frac{(-1+T_1)T_1(-1+T_{16})}{T_1^2-T_{16}-T_1T_{16}}$	$-\frac{(-1+T_1)^2(-1+T_{16})}{T_1^2-T_{16}-T_1T_{16}}$	0

```
Do[β = β // gm_{1,k→1}, {k, 11, 16}]; β
```

$$\begin{pmatrix} -\frac{1-4T_1+8T_1^2-11T_1^3+8T_1^4-4T_1^5+T_1^6}{T_1^3} & h_1 \\ t_1 & 0 \end{pmatrix}$$

```
<< KnotTheory`
Alexander[Knot[8, 17]][T1] // Factor
Loading KnotTheory` version of August 22, 2010, 13:36:57.55.
Read more at http://katlas.org/wiki/KnotTheory.
KnotTheory::loading : Loading precomputed data in PD4Knots`.
```

$$-\frac{1-4T_1+8T_1^2-11T_1^3+8T_1^4-4T_1^5+T_1^6}{T_1^3}$$

Where does it come from? The accidental[1] answer is that it is a symbolic calculus for a natural reduction[4] of the unique homomorphic expansion[2] of w-tangles[3].

1. "Accidental" for it's only how I came about it. There ought to be a better answer.
2. A "homomorphic expansion", aka as a homomorphic universal finite type invariant, is a completely canonical construct whose presence implies that the objects in questions are susceptible to study using graded algebra.
3. "v-Tangles" are the meta-group generated by crossings modulo Reidemeister moves. "w-Tangles" are a natural quotient of v-tangles. They are at least related and perhaps identical to a certain class of 1D/2D knots in 4D.
4. To "only what is visible by the 2D Lie algebra".

A certain generalization will arise by not reducing as in 4. A vast generalization may arise when homomorphic expansions for v-tangles are understood, a task likely equivalent to the Etingof-Kazhdan quantization of Lie bialgebras.

A Partial To Do List.

1. Where does it *more simply* come from?
2. Remove all the denominators.
3. How do determinants arise in this context (×2)?
4. Understand links.
5. Find the "reality condition".
6. Do some "Algebraic Knot Theory".
7. ~~Understand antipodes.~~
8. Categorify.

Montpellier w-pictures.